

[Chillout Zone](#) >

- [Start Here](#)
- [Top Tips](#)
- [Robots](#)
- [Something else](#)
- [Blogs](#)
- [Reviews](#)
- [Challenges](#)
- [Forums](#)
- [Recent](#)
- [About](#)
- [My Account](#)
- [My stuff](#)


[Home](#)**Arduino 101: Timers and Interrupts**

by [RobotFreak](#) Collected by 68 users
[timers](#), [timer](#), [pwm](#), [interrupt](#), [counter](#), [Arduino](#)
[Tip/walkthrough](#)



By [RobotFreak](#) @ August 7, 2011
 How to use timer and interrupts on Arduino boards.

- Programming language: [C++](#), [Arduino](#)



Attachment Size
[Arduino101-Timers.zip](#) 32.9 KB
[ArduinoTimer101.zip](#) 2.74 KB

Update 07/24/2013 04/24/2013

- Example 3 has been updated to work with Arduino v1.x.
- Added ArduinoTimer101.zip examples source code for Arduino v1.x.

This tutorial shows the use of timers and interrupts for Arduino boards. As Arduino programmer you will have used timers and interrupts without knowledge, because all the low level hardware stuff is hidden by the Arduino API. Many Arduino functions use timers, for example the time functions: [delay\(\)](#), [millis\(\)](#) and [micros\(\)](#) and [delayMicroseconds\(\)](#). The PWM functions [analogWrite\(\)](#) uses timers, as the [tone\(\)](#) and the [noTone\(\)](#) function does. Even the [Servo library](#) uses timers and interrupts.

What is a timer?

A timer or to be more precise a timer / counter is a piece of hardware built in the Arduino controller (other controllers have timer hardware, too). It is like a clock, and can be used to measure time events. The timer can be programmed with some special registers. You can configure the prescaler for the timer, or the mode of operation and many other things.

The controller of the Arduino is the Atmel AVR ATmega168 or the ATmega328. These chips are pin compatible and only differ in the size of internal memory. Both have 3 timers, called timer0, timer1 and timer2. Timer0 and timer2 are 8bit timer, where timer1 is a 16bit timer. The most important difference between 8bit and 16bit timer is the timer resolution. 8bits means 256 values where 16bit means 65536 values for higher resolution.

The controller for the Arduino Mega series is the Atmel AVR ATmega1280 or the ATmega2560. Also identical only differs in memory size. These controllers have 6 timers. Timer 0, timer1 and timer2 are identical to the ATmega168/328. The timer3, timer4 and timer5 are all 16bit timers, similar to timer1.

All timers depend on the system clock of your Arduino system. Normally the system clock is 16MHz, but for the Arduino Pro 3,3V it is 8MHz. So be careful when writing your own timer functions.

The timer hardware can be configured with some special timer registers. In the Arduino firmware all timers were configured to a 1kHz frequency and interrupts are generally enabled.

Timer0:

Timer0 is a 8bit timer.

In the Arduino world timer0 is used for the timer functions, like [delay\(\)](#), [millis\(\)](#) and [micros\(\)](#). If you change timer0 registers, this may influence the Arduino timer function. So you should know what you are doing.

Timer1:

Timer1 is a 16bit timer.

In the Arduino world the [Servo library](#) uses timer1 on Arduino Uno (timer5 on Arduino Mega).

Timer2:

Timer2 is a 8bit timer like timer0.

In the Arduino world the [tone\(\)](#) function uses timer2.

Timer3, Timer4, Timer5:

Timer 3,4,5 are only available on Arduino Mega boards. These timers are all 16bit timers.

Timer Register

You can change the Timer behaviour through the timer register. The most important timer registers are:

[Contact Us](#)**Search**

User login

Username: *

Password: *

- [Create new account](#)
- [Request new password](#)

[Log in using OpenID](#)

Recent blog posts

- [THE CD FILES #1](#)
- [Black Friday Treasure Hunt Sale Starts in 1 Week! Find Free Products and Discounts up to 99%](#)
- [Unique found object sculptures](#)
- [Why I want to build Companion Droids](#)
- [My Dream is becoming true?](#)
- [The Ultimate Guide To Guides On Making Star Wars Robots](#)
- [The Technology Showcase London 2016 with Makeblock](#)
- [Modelling a Servo Spline](#)
- [Planning my Robot](#)
- [Me, Robots, & Everything](#)

[more](#)**Bonus:**

- [Shops](#)
- [Cool Guides](#)
- [Other robot-related](#)


Latest weblinks

- [Pololu Robotics and Electronics Black Friday/Cyber Monday Sale 2016](#)
- [BEAM Robot World](#)
- [Airblock: The Modular and Programmable Starter Drone](#)
- [Project ORRCA](#)
- [Spot Mini](#)
- [Apple Home Automation](#)
- [R2-D2 plans and discussion](#)
- [Gloves that translate sign language to speech](#)
- [robot surgeon](#)
- [Some LMR bots doing standup and improv comedy...](#)

Navigation

- [LMR on Google+](#)
- [LMR on Facebook](#)
- [LMR on Twitter](#)
- [Creative Commons License Idea](#)
- [LMR Scrapbook](#)
- [User list](#)
- [Let's Make Robots World Maker Faire 2016](#)
- [Unread posts](#)
- [RobotShop Rover Development](#)
- [RSS feeds](#)
- [Spam Control](#)

TCCR_x - Timer/Counter Control Register. The prescaler can be configured here.



TCNT_x - Timer/Counter Register. The actual timer value is stored here.

OCR_x - Output Compare Register

ICR_x - Input Capture Register (only for 16bit timer)

TIMSK_x - Timer/Counter Interrupt Mask Register. To enable/disable timer interrupts.

TIFR_x - Timer/Counter Interrupt Flag Register. Indicates a pending timer interrupt.

Clock select and timer frequency

Different clock sources can be selected for each timer independently. To calculate the timer frequency (for example 2Hz using timer1) you will need:

1. CPU frequency 16Mhz for Arduino
2. maximum timer counter value (256 for 8bit, 65536 for 16bit timer)
3. Divide CPU frequency through the chosen prescaler (16000000 / 256 = 62500)
4. Divide result through the desired frequency (62500 / 2Hz = 31250)
5. Verify the result against the maximum timer counter value (31250 < 65536 success) if fail, choose bigger prescaler.

Table 16-5. Clock Select Bit Description

CS12	CS11	CS10	Description
0	0	0	No clock source (Timer/Counter stopped).
0	0	1	clk _{cpu} /1 (No prescaling)
0	1	0	clk _{cpu} /8 (From prescaler)
0	1	1	clk _{cpu} /64 (From prescaler)
1	0	0	clk _{cpu} /256 (From prescaler)
1	0	1	clk _{cpu} /1024 (From prescaler)
1	1	0	External clock source on T1 pin. Clock on falling edge.
1	1	1	External clock source on T1 pin. Clock on rising edge.

Timer modes

Timers can be configured in different modes.

PWM mode. Pulth width modulation mode. the OC_x outputs are used to generate PWM signals

CTC mode. Clear timer on compare match. When the timer counter reaches the compare match register, the timer will be cleared

Table 16-4. Waveform Generation Mode Bit Description⁽¹⁾

Mode	WGM13	WGM12 (CTC1)	WGM11 (PWM11)	WGM10 (PWM10)	Timer/Counter Mode of Operation	TOP	Update of OCR1x at	TOV1 Flag Set on
0	0	0	0	0	Normal	0xFFFF	Immediate	MAX
1	0	0	0	1	PWM, Phase Correct, 8-bit	0x00FF	TOP	BOTTOM
2	0	0	1	0	PWM, Phase Correct, 9-bit	0x01FF	TOP	BOTTOM
3	0	0	1	1	PWM, Phase Correct, 10-bit	0x03FF	TOP	BOTTOM
4	0	1	0	0	CTC	OCR1A	Immediate	MAX
5	0	1	0	1	Fast PWM, 8-bit	0x00FF	BOTTOM	TOP
6	0	1	1	0	Fast PWM, 9-bit	0x01FF	BOTTOM	TOP
7	0	1	1	1	Fast PWM, 10-bit	0x03FF	BOTTOM	TOP
8	1	0	0	0	PWM, Phase and Frequency Correct	ICR1	BOTTOM	BOTTOM
9	1	0	0	1	PWM, Phase and Frequency Correct	OCR1A	BOTTOM	BOTTOM
10	1	0	1	0	PWM, Phase Correct	ICR1	TOP	BOTTOM
11	1	0	1	1	PWM, Phase Correct	OCR1A	TOP	BOTTOM
12	1	1	0	0	CTC	ICR1	Immediate	MAX
13	1	1	0	1	(Reserved)	—	—	—
14	1	1	1	0	Fast PWM	ICR1	BOTTOM	TOP
15	1	1	1	1	Fast PWM	OCR1A	BOTTOM	TOP

What is an interrupt?

The program running on a controller is normally running sequentially instruction by instruction. An interrupt is an external event that interrupts the running program and runs a special interrupt service routine (ISR). After the ISR has been finished, the running program is continued with the next instruction. Instruction means a single machine instruction, not a line of C or C++ code.

Before an pending interrupt will be able to call a ISR the following conditions must be true:

- Interrupts must be generally enabled
- the according Interrupt mask must be enabled

Interrupts can generally enabled / disabled with the function `interrupts()` / `noInterrupts()`. By default in the Arduino firmware interrupts are enabled. Interrupt masks are enabled / disabled by setting / clearing bits in the Interrupt mask register (TIMSK_x).

When an interrupt occurs, a flag in the interrupt flag register (TIFR_x) is been set. This interrupt will be automatically cleared when entering the ISR or by manually clearing the bit in the interrupt flag register.

The Arduino functions `attachInterrupt()` and `detachInterrupt()` can only be used for external interrupt pins. These are different interrupt sources, not discussed here.

Timer interrupts

A timer can generate different types of interrupts. The register and bit definitions can be found in the processor data sheet ([Atmega328](#) or [Atmega2560](#)) and in the I/O definition header file (iomx8.h for Arduino, iomxx0_1.h for Arduino Mega in the hardware/tools/avr/include/avr folder). The suffix x stands for the timer number (0..5), the suffix y stands for the output number (A,B,C), for example TIMSK1 (timer1 interrupt mask register) or OCR2A (timer2 output compare register A).

Timer Overflow:

Timer overflow means the timer has reached its limit value. When a timer overflow interrupt occurs, the timer overflow bit TOV_x will be set in the interrupt flag register TIFR_x. When the timer overflow interrupt enable bit TOIE_x in the interrupt mask register TIMSK_x is set, the timer overflow interrupt service routine ISR(TIMEX_OVF_vect) will be called.

Output Compare Match:

When an output compare match interrupt occurs, the OCF_{xy} flag will be set in the interrupt flag register TIFR_x. When the output compare interrupt enable bit OCIE_{xy} in the interrupt mask register TIMSK_x is set, the output compare match interrupt service ISR(TIMEX_COMPy_vect) routine will be called.

Timer Input Capture:

When a timer input capture interrupt occurs, the input capture flag bit ICF_x will be set in the interrupt flag register TIFR_x. When the input capture interrupt enable bit ICIE_x in the interrupt mask register TIMSK_x is set, the timer input capture interrupt service routine ISR(TIMEX_CAPT_vect) will be called.

PWM and timer

There is a fixed relation between the timers and the PWM capable outputs. When you look in the data sheet or the pinout of the processor these PWM capable pins have names like OCR_xA, OCR_xB or OCR_xC (where x means the timer number 0..5). The PWM functionality is often shared with other pin functionality.

The Arduino has 3Timers and 6 PWM output pins. The relation between timers and PWM outputs is:

Pins 5 and 6: controlled by timer0

Pins 9 and 10: controlled by timer1
Pins 11 and 13: controlled by timer2

On the Arduino Mega we have 6 timers and 15 PWM outputs:

Pins 4 and 13: controlled by timer0
Pins 11 and 12: controlled by timer1
Pins 9 and 10: controlled by timer2
Pin 2, 3 and 5: controlled by timer 3
Pin 6, 7 and 8: controlled by timer 4
Pin 46, 45 and 44:: controlled by timer 5

Usefull 3rd party libraries

Some 3rd party libraries exists, that uses timers:

- [Ken Shirrifs IR library](#). Using timer2. Send and receive any kind of IR remote signals like RC5, RC6, Sony
- [Timer1 and Timer3 library](#). Using timer1 or timer3. The easy way to write your own timer interrupt service routines.

I have ported these libraries to different timers for Arduino Mega. All ported libraries can be found in the attached file.

Pitfalls

There exists some pitfalls you may encounter, when programming your Arduino and make use of functions or libraries that uses timers.

- Servo Library uses Timer1. You can't use PWM on Pin 9, 10 when you use the Servo Library on an Arduino. For Arduino Mega it is a bit more difficult. The timer needed depends on the number of servos. Each timer can handle 12 servos. For the first 12 servos timer 5 will be used (loosing PWM on Pin 44,45,46). For 24 Servos timer 5 and 1 will be used (loosing PWM on Pin 11,12,44,45,46).. For 36 servos timer 5, 1 and 3 will be used (loosing PWM on Pin 2,3,5,11,12,44,45,46).. For 48 servos all 16bit timers 5,1,3 and 4 will be used (loosing all PWM pins).
- Pin 11 has shared functionality PWM and MOSI. MOSI is needed for the SPI interface, You can't use PWM on Pin 11 and the SPI interface at the same time on Arduino. On the Arduino Mega the SPI pins are on different pins.
- tone() function uses at least timer2. You can't use PWM on Pin 3,11 when you use the tone() function an Arduino and Pin 9,10 on Arduino Mega.


Examples

Time for some examples.

Blinking LED with compare match interrupt

The first example uses the timer1 in CTC mode and the compare match interrupt to toggle a LED. The timer is configured for a frequency of 2Hz. The LED is toggled in the interrupt service routine.

Figure 16-6. CTC Mode, Timing Diagram



```
/* Arduino 101: timer and interrupts
1: Timer1 compare match interrupt example
more infos: http://www.letsmakerobots.com/node/28278
created by RobotFreak
*/

#define ledPin 13

void setup()
{
  pinMode(ledPin, OUTPUT);

  // initialize timer1
  noInterrupts(); // disable all interrupts
  TCCR1A = 0;
  TCCR1B = 0;
  TCNT1 = 0;

  OCR1A = 31250; // compare match register 16MHz/256/2Hz
  TCCR1B |= (1 << WGM12); // CTC mode
  TCCR1B |= (1 << CS12); // 256 prescaler
  TIMSK1 |= (1 << OCIE1A); // enable timer compare interrupt
  interrupts(); // enable all interrupts
}

ISR(TIMER1_COMPA_vect) // timer compare interrupt service routine
{
  digitalWrite(ledPin, digitalRead(ledPin) ^ 1); // toggle LED pin
}

void loop()
{
  // your program here...
}
```

Blinking LED with timer overflow interrupt

same example like before but now we use the timer overflow interrupt. In this case timer1 is running in normal mode.

The timer must be preloaded every time in the interrupt service routine.

```
/*
* Arduino 101: timer and interrupts
* 2: Timer1 overflow interrupt example
* more infos: http://www.letsmakerobots.com/node/28278
* created by RobotFreak
*/

#define ledPin 13

void setup()
{
  pinMode(ledPin, OUTPUT);

  // initialize timer1
  noInterrupts(); // disable all interrupts
  TCCR1A = 0;
  TCCR1B = 0;

  TCNT1 = 34286; // preload timer 65536-16MHz/256/2Hz
  TCCR1B |= (1 << CS12); // 256 prescaler
  TIMSK1 |= (1 << TOIE1); // enable timer overflow interrupt
  interrupts(); // enable all interrupts
}

ISR(TIMER1_OVF_vect) // interrupt service routine that wraps a user defined function supplied by attachInterrupt
{
  TCNT1 = 34286; // preload timer
  digitalWrite(ledPin, digitalRead(ledPin) ^ 1);
}


void loop()
{
  // your program here...
}
```

}

Reading quadrature encoders with a timer

The next example is part of my [Ardubot project](#). It uses timer2 and the compare match interrupt to read the encoder inputs. Timer2 is initialized by default to a frequency of 1kHz (1ms period). In the interrupt service routine the state of all encoder pins is read and a state machine is used to eliminate false readings. Using the timer interrupt is much easier to handle than using 4 input change interrupts.

The signals of a quadrature encoder is a 2bit [Gray code](#). Only 1 bit is changing from state to state. A state machine is perfect to check the signal and count the encoder ticks. The timer must be fast enough, to recognize each state change. For the [Pololu wheel encoders](#) used here, the 1ms timer is fast enough.



The following example has been modified to work with Arduino V1.x

```
/*
 *
 * Arduino 101: timer and interrupts
 * 3: Timer2 compare interrupt example. Quadrature Encoder
 * more infos: http://www.letsmakerobots.com/node/28278
 * created by RobotFreak
 *
 * Credits:
 * based on code from Peter Dannegger
 * http://www.mikrocontroller.net/articles/Drehgeber
 */

#if ARDUINO >= 100
#include "Arduino.h"
#else
#include "WConstants.h"
#endif

// Encoder Pins
#define encLtA 2
#define encLtB 3
#define encRtA 11
#define encRtB 12
#define ledPin 13

#define LT_PHASE_A          digitalRead(encLtA)
#define LT_PHASE_B          digitalRead(encLtB)
#define RT_PHASE_A          digitalRead(encRtA)
#define RT_PHASE_B          digitalRead(encRtB)

static volatile int8_t encDeltaLt, encDeltaRt;
static int8_t lastLt, lastRt;
int encLt, encRt;

ISR( TIMER2_COMPA_vect )
{
    int8_t val, diff;

    digitalWrite(ledPin, HIGH); // toggle LED pin
    val = 0;
    if( LT_PHASE_A )
        val = 3;
    if( LT_PHASE_B )
        val ^= 1;
    diff = lastLt - val;
    if( diff & 1 ){
        lastLt = val;
        encDeltaLt += (diff & 2) - 1;
    }

    val = 0;
    if( RT_PHASE_A )
        val = 3;
    if( RT_PHASE_B )
        val ^= 1;
    diff = lastRt - val;
    if( diff & 1 ){
        lastRt = val;
        encDeltaRt += (diff & 2) - 1;
    }
    digitalWrite(ledPin, LOW); // toggle LED pin
}

void QuadratureEncoderInit(void)
{
    int8_t val;

    cli();
    TIMSK2 |= (1<<OCIE2A);
    sei();
    pinMode(encLtA, INPUT);
    pinMode(encRtA, INPUT);
    pinMode(encLtB, INPUT);
    pinMode(encRtB, INPUT);

    val=0;
    if( LT_PHASE_A )
        val = 3;
    if( LT_PHASE_B )
        val ^= 1;
    lastLt = val;
    encDeltaLt = 0;

    val=0;
    if( RT_PHASE_A )
        val = 3;
    if( RT_PHASE_B )
        val ^= 1;
    lastRt = val;
    encDeltaRt = 0;

    encLt = 0;
    encRt = 0;
}

int8_t QuadratureEncoderReadLt( void ) // read single step encoders
{
    int8_t val;

    cli();
    val = encDeltaLt;
    encDeltaLt = 0;
    sei();
    return val;
}

int8_t QuadratureEncoderReadRt( void ) // read single step encoders
{
    int8_t val;
```

```

cli();
val = encDeltaRt;
encDeltaRt = 0;
sei();
return val; // counts since last call
}

void setup()
{
  Serial.begin(38400);
  pinMode(ledPin, OUTPUT);
  QuadratureEncoderInit();
}

void loop()
{
  encLt += QuadratureEncoderReadLt();
  encRt += QuadratureEncoderReadRt();
  Serial.print("Lt: ");
  Serial.print(encLt, DEC);
  Serial.print(" Rt: ");
  Serial.println(encRt, DEC);
  delay(1000);
}

```

Update 02/11/2014:

Q: Why have some variables the volatile keyword, like encodeDeltaLt and encodeDeltaRt in the examples above?

A: Read about using the volatile keyword here: <http://arduino.cc/en/Reference/Volatile>

Comment viewing options

Threaded list - expanded | Date - oldest first | 10 comments per page | [Save settings](#)

Select your preferred way to display the comments and click "Save settings" to activate your changes.

By [Gonzik](#) @ Mon, 2011-08-15 17:16



[Great tutorial RobotFreak!](#)

Great tutorial RobotFreak! Great reminder for those who know and comprehensive enough to teach newbies. Well done :)

By [Korel](#) @ Mon, 2011-08-15 18:57



[Thank you for this great](#)

Thank you for this great tutorial, it is well appreciated :)

By [hardmouse](#) @ Mon, 2011-08-15 21:41



[Thank you RobotFreak~~ This](#)

Thank you RobotFreak~~ This is helping me and so as many others~

By [RobotFreak](#) @ Tue, 2011-08-16 16:11



[Thank you all. Hope this](#)

Thank you all. Hope this tutorial is helpful for somebody.

By [isotope](#) @ Thu, 2011-08-18 02:39



[Thank you so much for taking](#)

Thank you so much for taking time and writing this!

By [westony](#) @ Sat, 2011-11-05 17:35



[Thank you so much about this](#)

Thank you so much about this tutorial. (BOW)

By [Cimbotic](#) @ Wed, 2012-01-18 12:46



[Awesome tutorial!!](#)

Man.. I wish I saw this when I was trying to figure out timer interrupts! This explains it much better than anything else I've seen online.

By [TernyKing](#) @ Mon, 2013-02-04 13:15



[May I link / abstract this for the http://ArduinoInfo.Info WIKI?](#)

Hi RobotFreak,

This is really good stuff, that is better than what I have right now on this subject.

May I link / abstract this for the <http://ArduinoInfo.Info> WIKI?

Regards, Terry King

...In The Woods in Vermont, USA

terry@yourduino.com

By [RobotFreak](#) @ Mon, 2013-02-04 16:43



[Hi Terry,thank you for](#)

Hi Terry,

thank you for asking. Feel free to link / abstract to this tutorial

Regards Peter

By [dishimwe](#) @ Sun, 2013-03-17 22:48



[Arduino Due Timer Interrupt](#)

Do you know how to do that on the Arduino Due?

[1](#) [2](#) [3](#) [4](#) [next »](#) [last »](#)

Let's make robots!

© 2016 RobotShop inc. All rights reserved.

[Terms & Conditions](#)